"FILE SEARCH"
IN ASTRO BASIC
BY MICHAEL WHITE

"WHAT IS FILE SEARCH?" SOME OUT THERE MAY ASK. WELL, IN COMPUTER "LINGO", A FILE IS A MEMORY DEVICE THAT STORES INFO IN SEQUENTIAL SEGMENTS. IT'S LIKE A CARD FILE IN A LIBRARY. EACH CARD CONTAINS INFORMATION ON WHERE TO FIND A PARTICULAR BOOK. ON A LONG TAPE, THE "FILE" SYSTEM TELLS THE COMPUTER WHERE TO FIND A PARTICULAR PROGRAM AND/OR DATA.

MOST COMPUTER SYSTEMS HAVE A BUILT-IN FILE SEARCH THAT CAN "CALL UP" A PROGRAM, ETC., BY INPUTTING A CODE WORD OR NUMBER. THIS WORD OR NUMBER IS STORED ON THE DISK OR TAPE IMMEDIATELY BEFORE THE DESIRED DATA. THE COMPUTER THEN RUNS THE TAPE OR DISK SEARCHING FOR THE CODE. WHEN THE CORRECT CODE IS FOUND, THE PROGRAM FOLLOWING WILL LOAD.

FIVE YEARS AGO, WHEN THE OLD BALLY BASIC FIRST CAME OUT, THIS WAS AN INCLUDED FEATURE. ON PAGE 4 VOLUME 3 OF THE ARCADIAN IS THE TUTORIAL THAT EXPLAINS WHAT IT IS AND HOW TO USE IT. NOW, I'M NOT GOING TO RESTATE ANYTHING HERE. IF YOU'RE UNFAMILIAR WITH THE OLD SYSTEM, REREAD VOLUME 3. BACK ISSUES ARE AVAILABLE!

THE OLD BALLY BASIC FILE SEARCH, HOWEVER, HAD THREE SERIOUS FLAWS. THE FIRST BEING THE FACT THAT THE INTERFACE WOULD 'OPEN' PREMATURELY! FOR EXAMPLE, TO GET TO SAY PROGRAM #4 ON YOUR TAPE, YOU WOULD USE [:INPUT 4] AND PLAY THE TAPE. THE INTERFACE 'CLOSES' AND WON'T LOAD DATA UNTIL THE CODE IS FOUND (THE TWO BYTE SEQUENCE OF 0 AND 4). IF THAT SEQUENCE APPEARS ANYWHERE IN PROGRAMS #1, 2, OR 3 (THE PROGRAMS THAT ARE TAPED AHEAD OF #04), THE INTERFACE WILL 'OPEN' AND TRY TO LOAD DATA! THIS DOESN'T OCCUR VERY OFTEN IF YOUR PROGRAMS WERE SIMPLE AND TAPED USING [:PRINT ;LIST], BUT GETS WORSE IF REM STATEMENTS OR DATA STRINGS ARE INVOLVED, SUCH AS WHAT MIGHT BE IN A PROGRAM USING THE DAVE IBACH TAPING METHOD ON PAGE 24-25 OF THE ARCADIAN VOLUME 3. WORSE YET IS EXPERIENCED IF THE SCREEN PICTURE GETS TAPED WITH THE PROGRAM (SIMILAR TO ASTRO BASIC) USING THE METHOD THAT THE PROGRAM "SCRIBBLE AND RECORD" HAS!

IF YOU THINK THIS ISN'T POSSIBLE, LET ME SAY, IT IS! "SCRIBBLE AND RECORD" WAS ONE OF ABOUT 5 SAMPLE PROGRAMS THAT CAME WITH THE **Z80 MINICOURSE** (SEE VOLUME 4, PAGE 69 ARCADIAN). THE SCREEN TAPING ROUTINE IS ACCOMPLISHED IN **BASIC**, DESPITE THE NATURE OF THE COURSE ITSELF, WHICH IS Z80 MACHINE LANGUAGE. THIS ROUTINE LOOKS LIKE THE DAVE IBACH METHOD MENTIONED ABOVE! AS YOU ALL (SHOULD) KNOW, THE SCREEN (PICTURE, BASIC TEXT, AND ALL) DOESN'T EXHIST AT %(-24576), ONLY THE TEXT DOES, (AND ONLY IN ASTRO BASIC OR OLD BALLY BASIC). THE SCREEN STARTS AT %(16384)! SO WE SIMPLY CHANGE ADDRESSES AND ACCOMPLISH THE FEAT! TO TRY IT YOURSELF, PUT THIS INTO A PROGRAM SOMEWHERE:

```
)(XX)A=16384;B=%(20050);C=19999;CLEAR ;NT=1;:PRINT ;PRINT "FOR A=16384TO 19999
     ;%(A)=KP;NEXT A;%(20050)=",#6,B,";:RETURN ;RUN ";:RETURN
)(YY)CLEAR ;(YOUR SCREEN PICTURE DRAWING ROUTINE)
)(ZZ)NT=0;:PRINT ;FOR B=ATO C;CY=-39;CX=-77;TV=%(B);NEXT B;:RETURN
```

THE (XX), (YY), AND (ZZ) ABOVE, MEAN, USE ANY LINE NUMBERS YOU WISH. ALSO, YOU MAY USE OTHER VARIABLES FOR A, B, AND C, BUT DON'T 'REUSE' THOSE SAME 3 ELSEWHERE! ALSO, THE LOWER LEFT CORNER OF THE SCREEN WILL GET MESSED UP A BIT! (CAN'T BE AVOIDED, SORRY ABOUT THAT). BE SURE YOUR TAPE IS RECORDING BEFORE YOU START THIS AND LEAVE 5 MINUTES OF RECORDING TIME! USE:

```
):INPUT (TO RELOAD)
```

AS YOU CAN IMAGINE, THE BYTES GOING TO THE TAPE USING THIS TECHNIQUE CAN BE ANY NUMBER! IF THE INTERFACE 'OPENS' DURING A FILE SEARCH HERE, THE SCREEN FILLS UP WITH DATA, EVEN SCROLLS! CHANGING CODE NUMBERS IS NO SOLUTION EITHER. THE [:INPUT 4] (OR WHATEVER) COMMAND WOULD HAVE TO BE REINSTATED FOR EVERY SINGLE PROGRAM! IT CAN BE DONE, BUT IS A NUISANCE AND LOOKS BAD AT BEST! IF AT THIS POINT, THE SEQUENCE ACTUALLY RESEMBLES BASIC TEXT, IT JUST MIGHT EVEN LOAD, AND CAUSE A PROGRAM CRASH!

AND IF THAT WASN'T BAD ENOUGH, THE NEXT HURDLE WAS FAR WORSE! IT TAKES APPROXIMATELY 3 TO 4 MINUTES TO LOAD AN AVERAGE PROGRAM IN OLD BALLY BASIC! THAT MEANS TO REACH SAY, PROGRAM #10 ON YOUR TAPE, YOU COULD START IT RUNNING, ORDER A PIZZA, LEAVE HOME, WAIT FOR IT, PICK IT UP, RETURN HOME, AND HAVE IT HALF EATEN AS YOUR PROGRAM STARTS TO LOAD **30 TO 40 MINUTES LATER**!!! THIS REASON **ALONE** STOPPED ABOUT EVERYONE COLD WHEN IT CAME TO FILING PROGRAMS ON TAPE. (IN OLD BALLY BASIC)

WHEN THE NEW ASTRO BASIC CAME OUT, THE FILE SEARCH, ALONG WITH A FEW OTHER ROUTINES WERE TAKEN OUT TO MAKE ROOM FOR SOME NEW FEATURES. (EDITING, *() ARRAY, 2000 BAUD, AND OTHER GOODIES). GETTING THE EDITOR AND 2000 BAUD TAPING ALONE WAS WORTH WHATEVER IT COST TO MAKE THE CHANGE, BUT NOW THAT WE HAD A MACHINE THAT COULD TAPE FAST ENOUGH TO EFFICENTLY USE FILE SEARCHING, WE NO LONGER HAD IT!!! BUT THAT NEVER STOPS A TRUE PIONEER AND/OR INVENTOR. WHO WAS IT THAT SAID, "IF IT CAN BE DONE, IT HAS! IF IT CAN'T, IT WILL BE!"? FIRST SCREEN TAPING IN OLD BALLY BASIC (BB), THEN AUTO-RUN IN ASTRO BASIC (AB), AND NOW FILE SEARCH IN AB! WHERE WILL IT GO NEXT? EVERY DAY THE WORD "IMPOSSIBLE" GETS A NEW DEFINITION.

THE OTHER PROBLEM WITH FILE SEARCH IN OLD BB IS ALSO IN AB. AND THAT IS, TAPING QUALITY BECOMES CRITICAL! IN BB A FILED TAPE USUALLY TENDS TO FILL UP, SO THE USER IS TEMPTED TO USE LONGER TAPES. BUT IF YOU DO, YOU RUN THE RISK OF TAPE STRETCHING. WHEN THE RECORDER REWINDS IT COMES TO THE END OF THE TAPE SUDDENLY! **BANG**! THIS HAMMERING STRETCHES THE END OF THE TAPE!

RULE ONE OF TAPING, DON'T USE THE FIRST FEW SECONDS, AS TEMPTING AS IT MIGHT BE, AND CHECK FOR A LEADER! SOME (THE MORE EXPENSIVE, USUALLY) TAPES HAVE A SHORT PIECE ON EITHER END TO PROTECT THE ENDS FROM HAMMERING! THESE "LEADER TAPES" DO NOT RECORD AND HAVE TO BE BYPASSED.

RULE TWO, **NEVER** USE TAPES LONGER THAN C90 FOR ANYTHING, AND C60 OR LESS FOR THE BALLY. THE REASON IS THICKNESS! A C60 IS .75 MIL. MYLAR BASE A C90 IS .5 MIL., ANYTHING LONGER IS THAT MUCH THINNER. THIS MAKES THE TAPE WAY ,TOO, DELICATE TO AVOID STRETCHED SPOTS, OR EVEN BREAKAGES! ALSO, IT ALLOWS "BLEED-THROUGH". THIS IS WHERE ONE LAYER OF TAPE SLOWLY MAGNETIZES ANOTHER BECAUSE OF PROXIMITY. YOU CAN HEAR IT IN MUSIC TAPES, ON THE BALLY IT'S AN EXCEDRIN NUMBER!

RULE THREE IS TO SPEND A FEW PENNIES AND GET QUALITY TAPES! USING BARGAN BASEMENT TAPES IS BAD ENOUGH FOR MUSIC, AND HARD ON RECORD HEADS BECAUSE OF INFERIOR LUBRICANT CAUSING EXCESSIVE WEAR, BUT TRYING TO USE THEM FOR COMPUTER PROGRAMS IS A NIGHTMARE, AND FOR FILE SEARCHING, NEARLY IMPOSSIBLE! I CAN'T EVEN DESCRIBE THE HORRORS! A GOOD QUALITY "MASTER" TAPE IS MADE SO AS NOT TO HAVE, BLEED-THROUGH, DROP-OUTS, HISS, STRETCHING, WARPING, SQUEELING, JAMMING, SLIPPING, BREAKING, WOW AND FLUTTER, OR ANY OTHER ANOMALYS. BELIEVE ME, YOU'LL BE GLAD YOU SPENT THE LITTLE BIT EXTRA, OR BE SORRY IF YOU DIDN'T.

HOWEVER, THE OPPOSITE IS TRUE WHEN SELECTING A RECORDER. BIG STEREO MACHINES WITH DOLBY, DBX, ALC, ATC, ANL, AND OTHER "GOODIES", ONLY PROVE TO BE BAD NEWS FOR THE BALLY! THE COMPUTER WANTS TO HEAR [:INPUT] THE SAME TONES (AND ETC.) THAT IT SENT OUT! [:PRINT ;LIST] (OR WHATEVER). ALL THESE FANCY ADD-ON'S ONLY DISTORT THE ORIGIONAL SIGNAL SO AS TO ENHANSE THE SOUND FROM A SMALL SPEAKER TO THE HUMAN EAR. UNFORTUNATELY BALLY IS NOT HUMAN, AND HEARS THESE ENHANCEMENTS AS PURE DISTORTION! HENSEFORTH, THE SIMPLEST AND CHEEPEST RECORDERS WORK BEST! ONE WITH ONLY A VOLUME AND TONE KNOB IS OK, ANYTHING MORE AND YOU'RE FIGHTING IT. I USE A PAIR OF CTR-51'S FROM RADIO SHACK, REPLACING MY OLD SONY STEREO PORTABLE, AND HAVE NOTICED THE IMPROVEMENT! THIS NEWSLETTER HAS RECOMMENDED THE PANASONIC "SLIM LINE" AS ABOUT THE BEST THEY COULD FIND!

ALSO, THERE IS A PROCEDURE THAT WHEN USED, IMPROVES THE QUALITY OF 2000 BAUD FILE SEARCHED TAPES IN PARTICULAR! IT'S CALLED "QUEUING", AND WORKS BY ELIMINATING ONE OF TWO 'GLITCHES' ON THE TAPE, AND HIDING THE OTHER. THERE IS A CIRCUIT IN TAPE RECORDERS CALLED A BIAS OSCILLATOR. THIS COMES ON DURING RECORD MODE (NOT ON FOR PLAYBACK) AND SUPPLIES THE ERASE AND RECORD HEADS WITH A 100,000 CYCLE SIGNAL. THIS SIGNAL ERASES THE TAPE BY BREAKING THE MAGNETIC RESISTANCE OF THE METAL OXIDE (OR WHATEVER THEY NOW USE, BROWN PART OF TAPE), WHICH IS KNOWN AS A "HISTORIESES LOOP" WHEN CHARTED. THIS ONLY BECOMES AUDIBLE WHEN THE TAPE STARTS MOVING (RECORDING) 'START GLITCH', OR STOPS 'STOP GLITCH'. THEN A POP OR SNAP MAY SOMETIMES BE HEARD.

TO MINIMISE THE EFFECT OF 'GLITCHES' WHEN RECORDING BALLY PROGRAMS, LET THE RECORDER RUN ON OUT ABOUT 5 SEC. AFTER THE CURSOR RETURNS WHEN TAPING. THEN DISCONNECT THE RECORDER, TURN THE VOLUME **WAY DOWN** (OR RISK YOUR EARS), REWIND A BIT, AND PLAY IT BACK THROUGH THE SPEAKER, STOPPING THE TAPE JUST AS THE TONE ENDS. TAPING THE FOLLOWING PROGRAM FROM HERE ERASES THE 'STOP GLITCH' WHICH WAS 5 SECONDS FARTHER DOWN THE TAPE, AND PLACES THE 'START GLITCH' RIGHT TIGHT AGAINST THE PRECEDING PROGRAM SO OUR FILE SEARCHER CAN "STEP OVER" IT!

YOU SEE, FILE SEARCH IN AB USES A DIFFERENT PRINCIPLE THAN THAT OF OLD BB. BB LOOKED AT EVERY BYTE COMMING IN FROM THE TAPE. AB LOOKS AT ONE VALUE, (TWO BYTES), THEN "STEPS OVER" ANY UNWANTED DATA. THE KEY TO THIS FEAT IS TIMING. AB PROGRAMS VARY IN LENGTH FROM 21 SEC. FOR A NORMAL [:PRINT] DUMP, TO 24 SEC. FOR A DUMP INCLUDING THE LINE INPUT BUFFER. (SELF START). NOT MUCH DIFFERENCE! SO IT IS EASY TO "STEP OVER" A PROGRAM ON TAPE. SIMPLY SET UP A DELAY LOOP TO TIE THE COMPUTER UP LONG ENOUGH FOR THE UNWANTED PROGRAM TO PASS. I HAVE FOUND THAT:

```
>FOR D=0TO 11000;NEXT D
```

TAKES ABOUT 25 SEC. TO EXECUTE. THIS GIVES ENOUGH TIME TO "STEP OVER" THE UNWANTED DATA AND ANY GLITCHES ON THE TAPE. NOW, THE ONLY PART LEFT IS TO TELL THE COMPUTER WHICH PROGRAM TO :INPUT AND WHICH TO "STEP OVER". THIS CAN BE DONE WITH AN "IF " STATEMENT. SINCE THIS USES TWO LINES, WE'LL NUMBER THEM 60 AND 70 FOR NOW, AND PUT THEM IN OUR "MENU" PROGRAM.

```
>60 IF Y=G:INPUT ;RUN
>70 FOR D=0TO 11000;NEXT D;GOTO 60
```

NOW, WHEN WE PLAY THE TAPE, THE MENU PROGRAM WILL ENTER FIRST, THEN WE STOP THE RECORDER AND A MENU IS SHOWN. WE THEN MAKE A SELECTION AND PRESS "PLAY" ON THE RECORDER. NOW, OUR "CODE" IS STORED IN THE VARIABLE "G". (SEE LINE # 60 ABOVE). THE VARIABLE "Y" WILL BE :INPUT FROM THE TAPE. IF THEY ARE THE SAME, OUR PROGRAM WILL LOAD AND RUN. IF NOT, THE LOOP STEPS OVER THE UNWANTED PROGRAM AND SETS UP TO TRY AGAIN.

TO INPUT THE VARIABLE FROM TAPE, WE MUST FIRST KNOW TWO THINGS: ":PRINT " CAN BE USED TO DUMP ANY LENGTH DATA RUN AT ANY LOCATION BY ADDING THE ADDRESS OF THE DATA IMMEDIATELY AFTER, THEN A COMMA, THEN THE # OF WORDS OF DATA TO DUMP. (1 WORD = 2 BYTES). AN EXAMPLE OF THIS WOULD BE: [:PRINT %(-24576),500]. IT THEN WOULD BE LOADED USING: [:INPUT %(-24576)]. USING THIS METHOD WE CAN DUMP A VERY SHORT BURST OF DATA AHEAD OF EACH OF OUR "SLAVE" PROGRAMS ON THE TAPE.

THE OTHER THING WE MUST KNOW IS WHERE OUR ADDRESSES ARE. THE ENTIRE LIST OF AB VARIABLES, WITH THIER RESPECTIVE ADDRESSES IS ON PAGE 59 OF VOL.5 (THIS VOLUME) OF THIS NEWSLETTER! (THE ARCADIAN). KNOWING ALL THIS, CHANGE LINE # 60 TO INCLUDE OUR ":INPUT " STATEMENT:

```
>60 :INPUT %(20016);IF Y=G:INPUT ;RUN
```

THIS WILL LOAD THE VARIABLES BEGINNING AT %(20016), INCLUDING "Y"! THEN COMPARE IT TO "G" (OUR DESIRED PROGRAM), AND REACT ACCORDINGLY. TO DUMP OUR "SLAVE" PROGRAMS ONTO THE TAPE, FIRST SET "Y" TO THE CODE, THEN USE:

```
:PRINT %(20016),18;:PRINT
```

LEAVE ABOUT 5-10 SECONDS BETWEEN PROGRAMS TO ALLOW FOR THE DELAY LOOP TO FINISH EACH TIME. (BEFORE PRESSING "GO").

IF YOU WISH TO PRESET "Y" IN ANY PROGRAM THAT GETS FILE SEARCHED THIS WAY SIMPLY CHANGE THE ABOVE TAPING COMMAND TO:

```
>Y=(CODE NO.);:PRINT %(20016),18;Y=(YOUR NO.);:PRINT
```

THE "BURST" IS TAPED WITH THE CODE NO. BUT YOUR PROGRAM HAS A DIFFERENT "Y" VALUE FROM TAPE THIS WAY. THE CODE NUMBERS ARE DETERMINED BY THE MENU PROGRAM AND HOW YOU SET IT UP. IN OUR MENU WE DISPLAY NUMBERS AND TITLES, THEN CHOOSE FROM THE KEYPAD, STORING OUR SELECTION IN VARIABLE "G". THESE LINES DO THAT:

```
>20 CLEAR ;FOR G=1TO %(20111)÷10-8;PRINT #3,G,;GOSUB 80
>30 NEXT G;PRINT "  YOUR CHOICE?
>40 G=KP-48;IF (G<1)+(G>%(20111)÷10-8);GOTO 40
```

LINES 20 & 30 PRINT OUR MENU. LINE 40 SELECTS OUR CHOICE AND CHECKS LIMITS. IN THIS MENU PROGRAM, A LOOP STARTING AT "1" IS WHAT DETERMINES OUR FILE SEARCH CODE. ALSO, BECAUSE THE KEYPAD INPUT (LINE 40) IS A SINGLE STROKE, IT STOPS AT "9". SO OUR LIMITS ARE "1 TO 9", WHICH FITS ON ONE SIDE OF A C10 CASSETTE (IN AB). ONE STRANGE TWIST IS POSSIBLE THOUGH. THE "SLAVE" PROGRAMS DON'T HAVE TO BE TAPED IN ANY PARTICULAR ORDER! MENU SELECTION #4 CAN BE AHEAD OF #3 ON THE TAPE, IF IT'S NOT ,TOO, CONFUSING TO SOMEONE EXPECTING EVERYTHING TO COME UP IN THE ORDER SHOWN IN THE MENU. SUBROUTINE 80 PRINTS OUR TITLES, AND LOOKS LIKE THIS:

```
>80 IF G=1CX=-(# OF CHARACTORS AND SPACES-1)x3;PRINT "(YOUR TITLE #1)
>90 THRU >170 (SAME AS LINE #80 FOR TITLE #2 ETC.)
>(LAST LINE NO.) RETURN
```

THE FORMULA SHOWN TO FIGURE "CX" WILL CENTER OUR TITLES HORIZONTALLY ON THE SCREEN AND GIVE IT A SEMI-PROFESSIONAL LOOK. THE (LAST LINE NO.) NEEDS TO BE 10 HIGHER THAN THE LINE NUMBER USED FOR YOUR LAST TITLE. SO IF YOUR TAPE HAS SAY, 6 "SLAVE" PROGRAMS, THE (LAST LINE NO.) WOULD BE 140. IF 9 PROGRAMS ARE FOLLOWING THE MENU, THE (LAST LINE NO.) WOULD BE 170. IF ONLY 2, (LAST LINE NO.) IS 100. ALSO, THIS "LAST LINE" MUST BE THE FINAL ENTRY BEFORE TAPING, AS IT CONTROLS %(20111)! TO FULLY UNDERSTAND THIS FUNCTION OF BASIC, A DEMONSTRATION IS CALLED FOR. PUT AB IN THE SLOT, HIT "RESET", AND KEY IN:

```
>PRINT %(20111)
```

WHAT DID YOU GET? (I GOT A 0 (ZERO).) HIT THE KEY SEQUENCE "WORDS","GO". WHAT LINE # ARE YOU AT? (I,M AT 10 (TEN).) NOW TELL BALLY TO PRINT YOUR NAME BY FINISHING LINE 10 LIKE SO:

```
>10 PRINT "(YOUR NAME)
```

RUN THE PROGRAM. IT SHOULD PRINT YOUR NAME. NOW KEY IN (WITHOUT A "RESET"):

```
>PRINT %(20111)
```

WHAT DID YOU GET NOW? (I GOT 10 (TEN).) HIT THE KEY SEQUENCE "WORDS","GO" AGAIN. WHAT LINE # ARE YOU AT NOW? (I'M AT 20 (TWENTY).) FINISH THE LINE THUS:

```
>20 PRINT "IS COOL!
```

NOW RUN THE PROGRAM AND COOL OUT! AGAIN, (WITHOUT RESETTING), KEY IN THE "WORDS","GO" SEQUENCE ! WHAT DID YOU GET THIS TIME? (I GOT 30 (THIRTY).) HIT "RESET", AND KEY IN:

```
>%(20111)=9990
```

NOW TRY THE SEQUENCE "WORDS","GO"! WHAT DID YOU GET THIS TIME? (I GOT 10000 (TEN THOUSAND) BELIEVE IT OR NOT!) ARE YOU NOW FAMILIAR WITH THE +10 ADVANCE REGISTER? IN OLD BB IT WAS AT %(20068). HOWEVER, IN AB, IT GOES TO TAPE WITH THE PROGRAM! SO IF WE WERE TO ADD A NEW ENTRY TO THE END OF OUR MENU, JUST KEY IN OVER THE OLD (LAST LINE NO.):

```
>(OLD LAST LINE NO.) IF G=(NEXT CODE NO.)CX=-(SAME AS ABOVE);PRINT "(NEW TITLE)
```

FINISH WITH THE KEY SEQUENCE: "WORDS","GO","WORDS","RETURN ","GO", AND TAPE OVER THE OLD MENU PROGRAM. ONE LAST THING, KEEP YOUR CODE NUMBERS IN SEQUENCE! DUPLICATE NUMBERS WILL LOOK FUNNY AND BE CONFUSING, AND A HOLE IN THE SEQUENCE MEANS THAT THE LOOPS WON'T EVEN PRINT ALL YOUR TITLES.

THE ONLY THING OUR MENU STILL NEEDS IS "PROMPTS", SO WE CAN FOLLOW ITS PROGRESS WHILE THE TAPE IS RUNNING! WITH THAT, OUR PROGRAM IS NOW COMPLETE:

FILE SEARCHER

BY MIKE WHITE

```
>10 CLEAR ;NT=0;CX=-24;CY=0;PRINT "ST0P TAPE
>20 FOR D=0TO 999;NEXT D;CLEAR ;FOR G=1TO %(20111)÷10-8;PRINT #3,G,;GOSUB 80
>30 NEXT G;PRINT "  YOUR CHOICE?
>40 G=KP-48;IF (G<1)+(G>%(20111)÷10-8);GOTO 40
>50 CLEAR ;CY=0;CX=-30;PRINT "START TAPE
>60 :INPUT %(20016);CLEAR ;CY=4;IF Y=G CX=-18;PRINT "L0ADING";GOSUB 80;
     :INPUT ;RUN
>70 CX=-36;PRINT "SEARCHING F0R";GOSUB 80;FOR D=0TO 11000;NEXT D;GOTO 60
>80 IF G=1CX=-(SEE ABOVE);PRINT "(TITLE #1)
>90 THRU >170 (ABOVE ALSO)
>(LAST LINE NO.) RETURN
```

TAPE IT WITH:

```
>CLEAR ;CX=(USE ABOVE FORMULA TO CALCALATE CX FOR CENTERING);CY=0;
     PRINT "(YOUR TITLE)";%(20156)=27195;:PRINT %(16384),1887
```

THIS GIVES OUR MENU A "SELF START" FEATURE. TO LOAD, REWIND THE TAPE TO THE BEGINNING, AND USE:

```
>:INPUT  (HIT "GO" AND PLAY THE TAPE! AFTER THAT, JUST FOLLOW THE "PROMPTS")
```

IF YOU WOULD LIKE TO SEE "FILE SEARCH" DEMONSTRATED TO THE FULLEST, SEND $15.95 TO:

MICHAEL D. WHITE
R.D.#1 BOX 258 GREEN RD.
WAKEMAN, OHIO 44889

AND ASK FOR "QUADRA"! YOU MAY BE SURPRISED!!!